

# Mikrocontroller-Grundlagen

## Teil 14

*In der vierzehnten Folge dieser Artikelserie betrachten wir die Betriebsarten 1, 2 und 3 der seriellen Schnittstelle der MCS51-Mikrocontroller-Familie.*

### 5.2 Seriell-Mode 1

Die im Mode 0 verwendete Datenübertragungsart benötigt in jeder Richtung mindestens 2 Leitungen (Takt und Daten). Die asynchrone Datenübertragung benötigt dagegen keine Taktleitung, setzt aber voraus, daß der Sender und der Empfänger mit dem gleichen Datenformat und der gleichen Datenübertragungsgeschwindigkeit arbeiten. Die Synchronisation von Sender und Empfänger erfolgt durch die übertragenen Daten selbst.

Abbildung 118 zeigt das Datenübertra-

gungsformat des MCS51-Mikrocontrollers in der Betriebsart 1. Mit der fallenden Flanke des Senders synchronisiert sich die serielle Schnittstelle des Empfängers. Dieser tastet die Empfangsleitung in der Bitmitte des Start-Bits ab.

Sollte zu diesem Zeitpunkt ein falscher Pegel vom Empfänger erkannt werden, so bricht dieser den Datenempfang ab. Durch diese Sicherheitsmaßnahme werden Störspikes, die eine Länge von bis zu der halben Baudrate erreichen, sicher unterdrückt. Die Abtastung der folgenden Datenbits erfolgt ebenfalls in der Bitmitte, um eine

hohe Störunterdrückung zu erreichen.

Beim Empfänger wird das Empfangsflag RI in der Mitte des folgenden Stopbits gesetzt, während der Sender bereits sein Sende-Flag TI unmittelbar nach dem Ausenden des letzten Datenbits setzt.

Die Datenübertragungsgeschwindigkeit (Baudrate) wird beim 8031/51 ausschließlich über die Timer1-Überlaufrate sowohl für den Sende- als auch Empfangsbetrieb bestimmt. Wie Abbildung 119 zeigt, läßt sich dieser Takt optional durch das Steuerbit SMOD durch 2 teilen. Der eigentliche Schiebetak für das Senden bzw. der Abtasttakt für das Empfangen wird anschließend noch durch 16 geteilt.

Das Steuerbit SMOD befindet sich in Bit 7 des nicht bit- aber direkt-adressierbaren Power-Control-Registers PCON, welches unter der Adresse 87H erreichbar ist.

Beim 8032/52 kann optional, wie in Abbildung 120 gezeigt, die Takterzeugung






Bild 118: Datenformat in der Betriebsart 1



**Bild 119: Takterzeugung für die serielle Datenübertragung in den Modi 1 und 3 (8031/51)**

pin liegende Information in sein Schieberegister übernimmt.

Nach dem Empfang des Stopbits wird dessen Pegel in das RB8-Bit des seriellen Port-Control-Registers SCON hineingeschrieben. Bei korrekter Datenübertragung muß hier im Modus 1 immer ein High-Bit empfangen werden.



**Bild 120: Takterzeugung für die serielle Datenübertragung in den Modi 1 und 3 (8032/52)**

für die serielle Datenübertragung auch vom Timer2-Überlauf generiert werden. Durch Setzen des sich in T2CON befindenden Steuerbits TCLK bzw. RCLK lassen sich die Übertragungsraten getrennt für den Sende- und Empfangsteil einstellen. Damit ist es ebenfalls möglich, für den Sende- und Empfangsteil unterschiedliche Übertragungsraten zu realisieren.

Die Baudrate bei Verwendung des Timer1-Überlaufs errechnet sich aus der nachfolgenden Formel:

$$\text{Baudrate} = \frac{2^{\text{SMOD}} \cdot \text{Zählerüberlaufrate}}{32}$$

Daraus läßt sich ableiten, daß bei gelöschtem Steuerbit SMOD = 0 die Baudrate sich aus der Zählerüberlaufrate geteilt durch 32 ergibt, während bei gesetztem Steuerbit SMOD = 1 die Baudrate sich aus der Zählerüberlaufrate geteilt durch 16 errechnet.

Abbildung 121 zeigt die grundsätzliche Funktion des seriellen Ports in der Betriebsart 1 während der Datenein- bzw. -ausgabe, die völlig unabhängig voneinander erfolgt. Die Taktsynchronisation erfolgt dabei wie beschrieben über den Timer1- bzw. Timer2-Überlauf.

Mit dem Schreiben der zu sendenden Daten in das Register SBUF beginnt die Hardware des Mikrocontrollers mit der asynchronen seriellen Datenausgabe. Dazu wird zunächst der „Teiler : 16“ zurückgesetzt und von der Ablaufsteuerung das Start-Bit generiert. Anschließend folgen die 8 Datenbits (zuerst LSB) und abschließend das Stop-Bit. Die Hardware setzt zu Beginn des Stop-Bits das Flag TI, welches je nach Freigabe einen Interrupt auslösen kann.


Der serielle Datenempfang beginnt mit der ersten fallenden Flanke an dem seriellen

len Daten-Eingangspin RxD. Die Ablaufsteuerung prüft nach Ablauf der halben Bit-Zeit (8 Zählerschritte) den Pegel des RxD-Bits.

Sollte dieser wieder auf High-Pegel liegen, so wird der Datenempfang abgebrochen. Anderenfalls setzt die Ablaufsteuerung den „Teiler : 16“ zurück, der nun mit jedem Überlauf die am seriellen Eingangs-

Mit Empfang des Stop-Bits wird gleichzeitig das Empfangsflag RI gesetzt, welches bei einer Interrupt-Freigabe eine Programm-Unterbrechung herbeiführen kann, um das empfangene Zeichen auszuwerten.

Abbildung 122 zeigt ein Testprogramm, welches nacheinander die ASCII-Zeichen „A“ bis „Z“ und die Zeilenumbruchssteuersignale Carriage Return und Line Feed



**Bild 121: Funktion des seriellen Ports in der Betriebsart 1**

Bild: 121

0335	PRGM41:			; Serielle Schnittstelle Mode 1 8Bit
0335				; Datenausgabe
0335 759840	L41:	MOV	SCON,#40H	; Mode 1, (8Bit Daten) Receive Disable
0338 758920		MOV	TMOD,#20H	; Timer 0 keine Funktion
0338				; Timer 1 Mode 2 C/T =0, Gate =0
033B 758780		MOV	PCON, #80H	; SMOD setzen, kein Teiler /2
033E 758DFA		MOV	TH1,#100H-6	; Nachladewert 6 = 11,0592 MHz/12/
033E				; 16/9600 Baud
0341 D28E		SETB	TR1	; Timer starten (TCON)
0343 D299		SETB	TI	; Sendeflag setzen, damit die
0343				; Datenausgabe beginnen kann
0345 7F41	L41SCHL:	MOV	R7, #'A'	; Mit „A“ beginnen
0347 EF	L41WDH:	MOV	A, R7	; Zeichen lesen
0348 715C		ACALL	L41AUSG	; Ausgabe des Zeichens
034A 0F		INC	R7	; nächstes Zeichen
034B EF		MOV	A, R7	; Zeichen lesen
034C 24A5		ADD	A, #0FFH -'Z'	; Vergleich vorbereiten
034E 50F7		JNC	L41WDH	; Schleife, wenn „Z“ noch nicht
034E				; erreicht
0350 740D		MOV	A, #00DH	; „Carriage Return“ laden
0352 715C		ACALL	L41AUSG	; Ausgabe des Zeichens
0354 740A		MOV	A, #00AH	; „Line Feed“ laden
0356 715C		ACALL	L41AUSG	; Ausgabe des Zeichens
0358 7164		ACALL	L41DEL	; Zeitverzögerung um 1 Sekunde
035A 80E9		SJMP	L41SCHL	; Schleife
035C	L41AUSG:			; Ausgabe des in A übergebenen
035C				; Zeichens auf die serielle Schnittstelle
035C 3099FD	L41AWDH:	JNB	TI L41AWDH	; Warten, bis der Sendepuffer leer ist
035F C299		CLR	TI	; Sendeflag löschen
0361 F599		MOV	SBUF, A	; Zeichen in den Sendepuffer
0361				; schreiben
0363 22		RET		; Zurück zum Hauptprogramm
0364	L41DEL:			; Verzögerung um eine Sekunde
0364 7A07		MOV	R2, #007	; 7 * 140 ms -> ca. 1 Sek.
0366 7800		MOV	R0, #00H	; 256 * 512 = 131072 Taktzyklen
0368 7900		MOV	R1, #00H	; Schleifenwert laden
036A D9FE	L41DEL1:	DJNZ	R1, L41DEL1	; innere Schleife 2 * 256 = 512 Zyklen
036C D8FC		DJNZ	R0, L41DEL1	; äußere Schleife
036E DAFA		DJNZ	R2, L41DEL1	; 7 mal
0370 22		RET		; zurück zum Hauptprogramm

**Bild 122: Testprogramm, welches nacheinander „A“ ..., „Z“ und CR LF überträgt**

(CR und LF) sendet.

Die im zweiten Teil dieser Artikelserie vorgestellte Mikrocontroller-Grundsaltung erlaubt es, mit Hilfe des RS232C/V24-Treibers eine Verbindung zwischen der Schaltung und einem PC herzustellen. Dazu ist der sich auf der Schaltung befindende Steckverbinder JP 2 mit Hilfe eines Jumpers zu schließen, und der PC mit einem 9poligen Verlängerungskabel (1 : 1) über seine serielle Schnittstelle anzuschließen.

Auf dem PC ist dann nach Laden des Betriebssystems ein Terminal-Programm zu starten, dessen Übertragungsparameter mit 9600 Baud, 8 Datenbits und einem Stopbit einzustellen sind, um so die übertragenen Zeichen direkt auf dem Bildschirm darzustellen.

Abbildung 123 zeigt die Weiterentwicklung des beschriebenen Programmes. Es empfängt ein vom PC gesendetes Datenbyte, gibt es an dem Port P1 aus, wo die LED-Ausgabeplatine anzuschließen ist und sendet das empfangene Zeichen zum PC zurück.

### 5.3 Seriell-Mode 2


In der Betriebsart 2 der seriellen Schnittstelle werden 9 Datenbits gesendet im Gegensatz zum Mode 1, wo 8-Bit-Nutzdaten übertragen werden.

Abbildung 124 zeigt das Datenübertragungsformat des MCS51-Mikrocontrollers in der Betriebsart 2. Beim Empfang der seriellen Information legt der Mikrocontroller das neunte empfangene Bit in dem bitadressierbaren Speicher RB8 ab, wo es für beliebige Verwendungszwecke zur Verfügung steht.


In umgekehrter Richtung wird beim Senden des seriellen Datenstroms der Inhalt des bitadressierbaren Registers TB8 als neuntes Datenbit gesendet. Zusammen mit

**Bild 123: Testprogramm, welches ein empfangenes Zeichen auf die LEDs an P 1 ausgibt und über die serielle Schnittstelle echot.**

0371	PRGM42:			; Serielle Schnittstelle Mode 1 8Bit
0371				; Datenein- und ausgabe
0371 759850	L42:	MOV	SCON,#50H	; Mode 1, (8Bit Daten) Receive Enable
0374 758920		MOV	TMOD,#20H	; Timer 0 keine Funktion
0374				; Timer 1 Mode 2 C/T =0, Gate =0
0377 758700		MOV	PCON, #00H	; SMOD =0, Teiler /2
037A 758DFD		MOV	TH1,#100H-3	; Nachladewert 3 = 11,0592 MHz/12/
037A				; 16/2/9600 Baud
037D D28E		SETB	TR1	; Timer starten (TCON)
037F D299		SETB	TI	; Sendeflag setzen, damit die
037F				; Datenausgabe beginnen kann
0381 759000		MOV	P1, #000H	; Alle LEDs aus
0384 718C	L42WDH:	ACALL	L42READ	; Zeichen von Ser. lesen
0386 F590		MOV	P1, A	; Ausgabe des Zeichens auf P1
0388 7194		ACALL	L42AUSG	; Ausgabe des Zeichens auf Ser.
038A 80F8		SJMP	L42WDH	; Schleife
038C	L42READ:			; Warten, bis ein Zeichen empfangen
038C				; und es dann lesen
038C 3098FD	L42RWDH:	JNB	RI L42RWDH	; Warten, bis Zeichen empfangen
038F C298		CLR	RI	; Empfangsflag löschen
0391 E599		MOV	A, SBUF	; Zeichen lesen
0393 22		RET		; zurück zum Hauptprogramm
0394	L42AUSG:			; Ausgabe des in A übergebenen
0394				; Zeichens auf die serielle Schnittstelle
0394 3099FD	L42AWDH:	JNB	TI L42AWDH	; Warten, bis der Sendepuffer leer ist
0397 C299		CLR	TI	; Sendeflag löschen
0399 F599		MOV	SBUF, A	; Zeichen in den Sendepuffer schreiben
039B 22		RET		; Zurück zum Hauptprogramm



**Bild 124: Datenformat in der Betriebsart 2**



**Bild 125: Takterzeugung für die serielle Datenübertragung in der Betriebsart 2**

dem Start- und Stop-Bit ergibt sich somit eine Übertragung von 11 Bits pro Byte. Verwendung findet das neunte Bit bei der Übertragung beispielsweise von Paritätsinformationen, die von der Steuersoftware zu berechnen bzw. auszuwerten sind.

Die Datenübertragungsgeschwindigkeit (Baudrate) wird nicht wie in der Betriebsart 1 vom Timer1- bzw. Timer2-Überlauf be-

stimmt, sondern ergibt sich ausschließlich aus der Oszillatorfrequenz. Wie Abbildung 125 zeigt, läßt sich optional noch ein Teilungsfaktor von 2 durch das Steuerbit SMOD bestimmen. Die Baudrate errechnet sich aus der nachfolgenden Formel:

$$\text{Baudrate} = \frac{2^{\text{SMOD}} \cdot \text{fosz}}{64}$$

Daraus läßt sich ableiten, daß bei gelöschtem Steuerbit SMOD = 0 die Baudrate sich aus der Oszillatorfrequenz geteilt durch 64 ergibt, während bei gesetztem Steuerbit SMOD die Baudrate sich aus der

Oszillatorfrequenz geteilt durch 32 errechnet.

Die Betriebsart 2 hat gegenüber den anderen den Vorteil, daß für die Baudratengenerierung kein Timer erforderlich ist, so daß diese für andere Verwendungszwe-

cke frei sind. Nachteilig ist allerdings, daß nur 2 Teilungsfaktoren zur Verfügung stehen.

#### 5.4 Seriell-Mode 3

Die Betriebsart 3 der seriellen Schnittstelle ist im Prinzip eine Kombination aus den Betriebsarten 1 und 2. Das Datenübertragungsformat ist wie in der Betriebsart 2 beschrieben 9 Datenbits zuzüglich Start- und Stopbit (Abbildung 124). Die Datenübertragungsrate ist wie in der Betriebsart 1 direkt vom Timer 1 bzw. Timer 2 überlaufabhängig und ergibt sich wie aus Abbildung 119 und 120 ersichtlich.

### 5.5 Multiprozessorsystem

Die Betriebsarten 2 und 3 übertragen wie beschrieben 9 Datenbits, wobei das neunte Bit von RB 8 aufgenommen wird. Anschließend folgt das Stop-Bit. Die serielle Schnittstelle kann so programmiert werden, daß nach dem Empfang des Stop-Bits ein Interrupt (Programmunterbrechung) des seriellen Ports nur dann ausgelöst wird, wenn RB8 gleich 1 ist. Die Steuerung der Funktion übernimmt das direkt adressierbare Steuerbit SM2 im Steuerregister SCON.

Verwendung findet diese spezielle Datenübertragungstechnik hauptsächlich in Multi-Prozessorsystemen, wo mehrere Prozessoren auf einen gemeinsamen Bus geschaltet sind (Netzwerk). Möchte ein am Bus angeschlossener Prozessor einen Datenblock an einen von mehreren Slaves übertragen, sendet er zunächst ein Adreß-

byte aus, das den betreffenden Slave adressiert. Dieses Adreßbyte unterscheidet sich dadurch von den Datenbytes, daß das neunte

Bit des Adreßbytes gesetzt ist, während es bei den Datenbytes 0 ist.

Ist jetzt das Steuerbit SM 2 gesetzt, so wird bei den Slaves nur ein Interrupt bei der Übertragung eines Adreßbytes ausgelöst. Damit kann jeder Slave-Prozessor das empfangene Adreßbyte daraufhin prüfen, ob er adressiert ist. Der adressierte Slave löscht sein Bit SM2 und trifft Vorbereitungen, um die nachfolgenden Datenbytes zu empfangen.

Die nicht adressierten Slave-Prozessoren lassen hingegen ihre Bits SM2 gesetzt und ignorieren die nachfolgend übertragenen Datenbytes. Nach erfolgter Datenübertragung setzt der Empfänger wieder sein Steuerbit SM 2. Bedingt durch die 8 übertragenen Bits ist somit eine Adressierung von maximal 256 Busteilnehmern möglich.

Tabelle 20 zeigt häufig verwendete, durch den Timer 1 erzeugte Baudraten, wobei durch Veränderung der Oszillatorfrequenz und der Teilungsfaktoren fast beliebig viele Variationsmöglichkeiten bestehen. Bei Veränderung der Oszillatorfrequenz ist natürlich zu beachten, daß sich damit natürlich auch die Anzahl der abgearbeiteten Befehle pro Zeiteinheit ändert.

Weitere Variationsmöglichkeiten bestehen darin, den beim 8032/52 verfügbaren zusätzlichen Timer 2 für die Baudratengenerierung einzusetzen.

Im nächsten Teil der Mikrocontroller-Grundlagen-Serie erläutern wir ausführlich die Interrupt-Struktur der MCS51-Familie.

## Die verschiedenen Betriebsarten der seriellen Schnittstelle eignen sich für unterschiedlichste Einsatzmöglichkeiten

**Tabelle 20: Häufig verwendete, durch den Timer1 erzeugte Baudraten**

Baudrate	Betriebsart des seriellen Ports	Oszillatorfrequenz	SMOD	Timer/Counter 1		
				C/T	Betriebsart	Reloadwert
1 MHz	0	12 MHz	x	x	x	x
375 kBaud	2	12 MHz	1	x	x	x
187,5 kBaud	2	12 MHz	0	x	x	x
57,6 kBaud	1,3	11,0592 MHz	1	0	2	0FFH
19,2 kBaud	1,3	12,0592 MHz	1	0	2	0FDH
9600 Baud	1,3	11,0592 MHz	0	0	2	0FDH
4800 Baud	1,3	11,0592 MHz	0	0	2	0FAH
2400 Baud	1,3	11,0592 MHz	0	0	2	0F4H
1200 Baud	1,3	11,0592 MHz	0	0	2	CE8H
600 Baud	1,3	11,0592 MHz	0	0	2	0D0H
300 Baud	1,3	11,0592 MHz	0	0	2	0A0H
150 Baud	1,3	11,0592 MHz	0	0	2	040H
110 Baud	1,3	6 MHz	0	0	2	072H